

ЛЕКЦИЯ 4

доц. д-р Стела Стефанова

Входен език на PSpice A/D. Описание на компоненти с модел и макромодел.

1. Структура на входния език на PSpice A/D

Входният език на PSpice A/D може да се раздели на две основни части, показани на Фиг. 1, които определят спецификата на симулатора:

- **Описание на компоненти** – правилата и синтаксиса за описание и обръщение към моделите и макромоделите на аналогови и цифрови компоненти;
- **Директиви (команди) за управление на заданието** – набор от команди и правила за дефиниране на различните видове анализи и техните параметри.




Фиг.1. Структура на входния език на PSpice A/D

2. Общ формат за описание на компоненти

При въвеждането на общия формат на описание на компоненти и команди са въведени следните правила за означение:

- [] – незадължителен параметър;
- < > - задължителен параметър;

















2.1. Име на компонент **1**




- Първият символ – задължително латинска буква;
- Типът на компонента се определя от резервираната първа буква, както е показано в Таблица 1.

2.2. Списък от възли **2**

- Изброяват се възли, между които е свързано устройството.
- Списъкът от възли се изброява в точно определен ред (Таблица 1).

Таблица 1. Означение на компонентите в PSpice A/D и ред на изброяване на техните възли

Първи символ	Компонент	Графично изображение	Ред на изброяване на възлите
B	GaAs FET (N-канален)		D G S
C	Кондензатор		+ -
D	Диод		A K
E	ИНУН (Източник на напрежение, управляван по напрежение)		+ -
F	ИТУТ (Източник на ток, управляван по ток)		+ -
G	ИТУН (Източник на ток, управляван по напрежение)		+ -
H	ИНУТ (Източник на напрежение, управляван по ток)		+ -
I	Независим източник на ток		+ -
J	Полеви транзистор с P канал		D G S
J	Полеви транзистор с N канал		D G S
K	Магнитопровод на трансформатор		+ -
L	Индуктивност		+ -
M	MOS транзистор с P канал		D G S подложка
M	MOS транзистор с N канал		D G S подложка
N	Ц/А интерфейс		
O	А/Ц интерфейс		
Q	Биполярен транзистор NPN		К Б Е [подложка]

Q	Биполярен транзистор PNP		К Б Е [подложка]
R	Резистор		
R_VAR	Потенциометър		
S	Ключ, управляван по напрежение	 S VOFF = 0.0V VON = 1.0V	+ -
T	Закъснителна линия		A ⁺ A ⁻ B ⁺ B ⁻
U	Цифрово устройство		
V	Описание на независим източник на напрежение		+ -
W	Ключ, управляван по ток	 W IOFF = 0.0mA ION = 1mA	+ -
X	Описание на макромодел		Редът на изброяване е такъв, какъвто е описан при създаване на макромодела с командата .SUBCKT

2.3. Име на модел 3

- незадължителен параметър – има случаи, когато този параметър липсва в описанието на компонента;
- При аналогови компоненти (резистори, кондензатори и бобини) обикновено се използват вградените модели на PSpice и този параметър може да бъде изпуснат;
- **Пример за използване на вграден модел:**
R12 1 2 10K.
- При полупроводниковите елементи името на модела е задължителен параметър, които се формира от резервиран символ и каталожен номер на прибора.
- **Пример за използване на модел:**
Q3 10 5 0 Q2N2220.

2.4. Числени стойности 4

2.4.1. Дефиниране на числени стойности на компонентите

Суфиксите за дефиниране на производни на основните мерни единици са дадени в Таблица 2.

Таблица 2. Производни на основните мерни единици

10 ⁻¹⁸ А (ато)	10 ³ К (кило)
10 ⁻¹⁵ F (фемто)	10 ⁶ МEG (мега)
10 ⁻¹² P (пико)	10 ⁹ G (гига)
10 ⁻⁹ N (нано)	10 ¹² T (тера)
10 ⁻⁶ U (микро)	
10 ⁻³ M (мили)	

2.4.1. Дефиниране на идентификатор на параметри

- **Идентификатор на параметри** – {аритметичен израз/параметър}.
- **Примери:**
V 5 0 {VIN}
I1 12 0 {2*PI*TIME*ID},
където параметрите са предварително дефинирани с командата **.PARAMS**
.PARAMS VIN=5 или **.PARAMS PI=3.14**.
- Изразът не може да се пренася от един на следващ ред. Ако той е по-дълъг от един ред е необходимо в идентификатора на параметри да се дефинира един параметър, който е описан с командата **.FUNCTION**
- например **.FUNCTION A=2*PI*TIME*ID**, като се приеме, че този израз е по-дълъг от един ред и после се използва например **I1 12 0 {A}**.

3. Формат за описание на компоненти с команда MODEL

Общият формат за описание на компоненти с команда MODEL се задава във вида:

```

1 .MODEL 2 <име на модела> [AKO: <име на модел прототип>]
3
+ <име на типа>
4
+ ( [<име на параметър> = <стойност> >
5
+ [<толеранс>=<стойност> ] ] ...
6
+ [ T_MEASURED = <стойност> ]
7
+ [ [ T_ABS = <стойност> ]
8
+ [ T_REL_GLOBAL = <стойност> ]
9
или
+ [ T_REL_LOCAL = <стойност> ] ] )
```

3.1. Име на модела **1**

- Започва с резервираната буква и следва каталожен номер на компонента;
- Примери: **D1N4001**, **Q2N2220**.

3.2. Име на модел-прототип **2**

[**AKO:** *<име на модел прототип>*]

AKO (A Kind Of)

- Текущият модел и моделът прототип трябва да бъдат от един и същи тип;
- В текущия модел се използват стойностите на всички параметри от модела прототип с изключение на тези параметри, които са дефинирани в текущия модел.

ПРИМЕР:

- модел прототип

.MODEL QDRIVE NPN (<списък параметри>)

- текущ модел

.MODEL QDRV_NEW AKO:QDRIVE NPN (BF=90)

Текущият модел и моделът прототип са дефинирани за **NPN** транзистор.

В текущия модел с име на модела **QDRIVE_NEW** се използват стойностите на всички параметри от модела прототип с име на модела **QDRIVE** с изключение на параметра **BF=90**, които е дефиниран в текущия модел.

3.3. Име на типа **3**

- Резервирана служебна дума, която указва точно типа на компонента;
- Използва се при описанието на моделите в библиотечните файлове;
- Служебните думи, използвани за дефиниране на типа на модела, са показани в Таблица 3.

Таблица 3. Служебни думи за дефиниране на името на типа на компонента

Тип на компонента (служебна дума)	Означение на компонента	Описание на компонента
CAP	Cxxx	Кондензатор
CORE	Kxxx	Магнитопровод на трансформатор
D	Dxxx	Диод
DINPUT	Nxxx	ЦАП
DOUTPUT	Oxxx	АЦП
GASFET	Vxxx	GaAs FET (N-канален)
IND	Lxxx	Индуктивност
ISWITCH	Wxxx	Ключ, управляван по ток
NJF	Jxxx	Полеви транзистор с N канал
NMOS	Mxxx	MOS транзистор с N канал
NPN	Qxxx	Биполярен транзистор NPN
PJF	Jxxx	Полеви транзистор с P канал
PMOS	Mxxx	MOS транзистор с P канал
PNP	Qxxx	Биполярен транзистор PNP
RES	Rxxx	Резистор
TRN	Txxx	Закъснителна линия
UADC	Uxxx	АЦП
UDAC	Uxxx	ЦАП
UDLY	Uxxx	Цифрова закъснителна линия
UEFF	Uxxx	Тригер (управляван по фронт)
UGFF	Uxxx	Тригер (управляван по ниво)
UGATE	Uxxx	Логически примитив
UIO	Uxxx	I/O модел на цифрово устройство
UTGATE	Uxxx	Лог. елемент в трето състояние
VSWITCH	Sxxx	Ключ, управляван по напрежение

3.4. Име на параметър и стойност **4**

- Задава се списък от параметри и техните стойности и ако този списък е празен, PSpice избира стойности на параметрите по подразбиране.

3.5. Толеранс **5**

- Задава се спецификация на случайно изменение на параметрите. Използва се за извършване на статистически (Monte Carlo) анализи, които се прилагат за изследване поведението на схемата при случайни изменения на параметри на моделите на компонентите, използвани в схемата.
- **DEV** – дефинира независимо (некорелирано) отклонение на параметрите на устройството, т.е. параметрите на всяко устройство се изменят независимо от параметрите на другите устройства.

- **LOT** – дефинира зависимо (корелирано) отклонение от параметрите на всички устройства, които използват този модел.

- **Пример:** . MODEL D1N4001 D(IS=12.10-9 DEV=+- 5%)

3.6. Температурни параметри на модела

3.6.1. Температура на измерване на параметрите

[T_MEASURED = <value>] **6**

- Незадължителен параметър;
- Задава температурата, при която са измерени параметрите, описани в командата .MODEL;
- Тази стойност замества номиналната температура по подразбиране, заложена в PSPICE = 27C°.

3.6.2. Физическа температура на устройството

[T_ABS = <value>] **7**

- Задава абсолютната (физическата) температура на разглежданото устройство.

$$t^{\circ}_{\text{device}} = T_ABS$$

3.6.3. Отклонение по отношение на текущата температура

[T_REL_GLOBAL = <value>] **8**

- Незадължителен параметър;
- Задава промяна по отношение на текущата (глобалната) температура, която се дефинира с команди .TEMP , или .STEP TEMP или .DC TEMP .

$$t^{\circ}_{\text{device}} = t^{\circ}_{\text{global}} + T_REL_GLOBAL$$

3.6.4. Отклонение по отношение на температурата на модел-прототип

[T_REL_LOCAL = <value>] **9**

- Незадължителен параметър;
- Задава промяна по отношение на температурата на модела прототип:

$$t^{\circ}_{\text{device}} = T_ABS(AKO \text{ model}) + T_REL_LOCAL$$

3.10. Примери за описание на компоненти с команда .MODEL

Пример за модел на биполярен транзистор от библиотека Bipolar.lib е показан на Фиг. 2.

```
.model Q2N2222 NPN(Is=14.34f Xti=3 Eg=1.11 Vaf=74.03 Bf=255.9 Ne=1.307
+   Ise=14.34f Ikf=.2847 Xtb=1.5 Br=6.092 Nc=2 Isc=0 Ikr=0 Rc=1
+   Cjc=7.306p Mjc=.3416 Vjc=.75 Fc=.5 Cje=22.01p Mje=.377 Vje=.75
+   Tr=46.91n Tf=411.1p Itf=.6 Vtf=1.7 Xtf=3 Rb=10)
*   National      pid=19              case=TO18
*   88-09-07 bam      creation
```

Фиг. 2. Модел на биполярен транзистор от библиотека Bipolar.lib

Пример за модел на диод транзистор от библиотека Diode.lib е показан на Фиг. 3.

```

.model D1N4001 D (Is=14.11n N=1.984 Rs=33.89m Ikf=94.81 Xti=3 Eg=1.11
+      Cjo=25.89p M=.44 Vj=.3245 Fc=.5 Bv=75 Ibv=10u Tt=5.7u)
*      Motorola
*      Semiconductor Databook (mid 1970s)
*      03 Jun 91 pwt creation

```

Фиг. 3. Модел на диод D1N4001 от библиотека Diode.lib

а) Дефиниране на моделни параметри на резистор

.MODEL RMAX RES (R=1.5 TC1=0.02 TC2=0.05)

Моделните параметри на резистор са дефинирани в кръглите скоби.

б) Дефиниране на толеранс на моделните параметри на диод

.MODEL DLOAD D (IS=1E-9 DEV 0.5% LOT 10%)

За моделния параметър **IS=1E-9** на диод са дефинирани независимо отклонение **DEV=0.5%** и зависимо (корелирано) отклонение **LOT=10%**.

в) Дефиниране на моделни параметри на транзистор на базата на модел-прототип

.MODEL QDRIV NPN (IS=1E-9 BF = 30)

.MODEL QDR-NEW AKO: QDRIV NPN (BF = 50 IKF = 50m)

Текущият QDR-NEW ще използва стойностите на всички параметри от модела прототип QDRIV (например параметъра **IS=1E-9**), с изключение на стойностите за параметрите, дефинирани в текущия модел - за параметъра BF ще се използва стойност = 50, а не тази от модела прототип =30, както и за параметър IKF ще се използва стойност = 50m.

г) Дефиниране на температурни моделни параметри на транзистор

.MODEL QNPN_NEW NPN (BF=30 T_ABS = 35 T_MEASURED = 0)

- Моделният параметър BF е измерен при температура 0 градуса;
- Всеки биполярен NPN транзистор, който се обръща към този модел ще има физическа температура на устройството 35 градуса;

д) Дефиниране на температурни моделни параметри на транзистор по отношение на глобалната температура

.TEMP 10 30 50

.MODEL Q_NEW NPN (моделни параметри T_REL_GLOBAL = -5)

- Първата команда дефинира глобалната температура (могат да се използват и другите команди за дефиниране на глобална температура **.DC TEMP**, **.STEP TEMP**) и предполага 3 стартирания на PSPICE за 10, 30, 50 градуса;

- Всеки биполярен транзистор, който се обръща към модела Q_NEW, се изследва за температури съответно 5, 25 и 45 градуса, защото е дефинирано отклонение - 5 градуса.

е) Дефиниране на температурни моделни параметри на транзистор по отношение на температурата на модел-прототип

.MODEL QNPN_PRO NPN (параметри T_ABS = 20)

.MODEL QNPN_NEW AKO: QNPN_PRO NPN (параметри T_REL_LOCAL=10)

- Първата команда дефинира температурата на модел-прототип;
- Температурата на всеки транзистор, обръщан се към модела **QNPN_NEW**, ще се определя от температурата на модела-прототип **T_ABS = 20** и от относителното изменение **T_REL_LOCAL = 10** т.е. температурата на транзистора ще бъде 30 градуса.

4. Описание на макромодел (подсхема) в PSpice A/D

На Фиг. 4 е показана в обобщен вид идеята за създаване на подсхема (макромодел) в PSpice A/D.




Фиг.4. Идеята за дефиниране на макромодел

4.1. Понятие за макромодел

- Отделен фрагмент от схемата или замесваща схема на компонент, за описанието на която се използват моделите на други компоненти;
- Операционните усилватели се описват с макромодел, тъй като за описание на вътрешната им структура се използват моделите на биполярни или полеви транзистори;
- Цифровите устройства се описват с макромодели, защото за създаването им се използват модели на логическите примитиви.


4.2. Общ формат за описание на макромодел

Структурата на създаване на макромодел е показана на Фиг. 5



Фиг. 5. Структура на описание на макромодел

- Команди за създаване на макромодел - **.SUBCKT** и **.ENDS**
- Общ формат на команда **.SUBCKT**:



- **1 Име на макромодел** – име на компонента, описван с макромодела;
- **2 Списък от възли** – ред на изброяването на възлите;
- **3 OPTIONAL** – ключова дума - служи за дефиниране на един или повече незадължителни възли в макромодела. Задава се име и стойност по подразбиране. Ако при обръщение към макромодела не са указани тези възли, то се използва тяхната стойност по подразбиране.

- Примери:

Изключително удобно при дефиниране на захранващи източници на цифрови устройства, например **DPWR= \$G_DPWR**. В случай, че нищо не е дефинирано за възела, се установява глобален символ за цифрово захранване \$G_DPWR.

- **4** **PARAMS** – ключова дума – дефиниране на списък от параметри и стойности, с които се извършва обръщението към макромодела.
- **Примери:**
MXTYMNPLY=0 задава времезакъсненията на цифровото устройство (по подразбиране, стойност = 0 означава *типични времезакъснения*);
IOLEVEL=0 – указва нивото на сложност на модела на А/Ц интерфейс (по подразбиране, стойност = 0 указва най-простия модел за А/Ц интерфейс)
- **5**. **TEXT** - ключова дума – за дефиниране на име на текстова променлива, предавана от описанието на основната схема към описанието на макромодела. Използва се само при моделиране на цифрови истройства.
- **Примери:** **TEXT: JED_FILE = "Prog.JED"**

4.3. Примери за макромодели

4.3.1. Макромодел на операционен усилвател

Макромоделът на операционен усилвател LM741 е показан на Фиг. 6 и е извлечен от библиотека **OPAMP.lib**

```

*-----
* connections:      non-inverting input
*                   | inverting input
*                   | | positive power supply
*                   | | | negative power supply
*                   | | | | output
*                   | | | | |
*                   | | | | |
*
.subckt uA741      1 2 3 4 5
*
c1  11 12 8.661E-12
c2  6 7 30.00E-12
dc  5 53 dy
de  54 5 dy
dlp 90 91 dx
dln 92 90 dx
dp  4 3 dx
egnd 99 0 poly(2),(3,0),(4,0) 0 .5 .5
fb  7 99 poly(5) vb vc ve vlp vln 0 10.61E6 -1E3 1E3 10E6 -10E6
ga  6 0 11 12 188.5E-6
gcm 0 6 10 99 5.961E-9
iee 10 4 dc 15.16E-6
hlim 90 0 vlim 1K
q1  11 2 13 qx
q2  12 1 14 qx
r2  6 9 100.0E3
rc1 3 11 5.305E3
rc2 3 12 5.305E3
re1 13 10 1.836E3
re2 14 10 1.836E3
ree 10 99 13.19E6
ro1 8 5 50
ro2 7 99 100
rp  3 4 18.16E3
vb  9 0 dc 0
vc  3 53 dc 1
ve  54 4 dc 1
vlim 7 8 dc 0
vlp 91 0 dc 40
vln 0 92 dc 40
.model dx D(Is=800.0E-18 Rs=1)
.model dy D(Is=800.00E-18 Rs=1m Cjo=10p)
.model qx NPN(Is=800.0E-18 Bf=93.75)
.ends

```

4.3.2. Макромодел на цифрова схема

Макромоделът на 4 битов брояч 74LS163 е показан на Фиг. 7. и е извлечен от библиотека **74LS.lib**

```
* 74LS163A Synchronous 4-bit Binary Counter with Direct Clear
*
* THE TTL LOGIC DATA BOOK, 1988, TI
* tc 06/30/92 Remodeled using LOGICEXP, PINDLY, & CONSTRAINT Devices
*
.SUBCKT 74LS163A CLK_I ENP_I ENT_I CLRBAR_I LOADBAR_I A_I B_I C_I D_I
+ QA_O QB_O QC_O QD_O RCO_O
+ OPTIONAL: DPWR=$G_DPWR DGND=$G_DGND
+ PARAMS: MNTYMXDLY=0 IO_LEVEL=0
*
U1 DFF(4) DPWR DGND $D_HI $D_HI CLK
+ DA DB DC DD QA QB QC QD QABAR QBBAR QCBAR QDBAR
+ D0_EFF IO_LS
*
U1S163ALOG LOGICEXP(17,15) DPWR DGND
+ CLK_I ENP_I ENT_I CLRBAR_I LOADBAR_I A_I B_I C_I D_I QA QB QC QD
+ QABAR QBBAR QCBAR QDBAR
+ CLK ENP ENT CLRBAR LOADBAR A B C D RCO DA DB DC DD IEN
+ D0_GATE IO_LS IO_LEVEL={IO_LEVEL}
+ LOGIC:
+ CLK = { CLK_I }
+ ENP = { ENP_I }
+ ENT = { ENT_I }
+ CLRBAR = { CLRBAR_I }
+ LOADBAR = { LOADBAR_I }
+ A = { A_I }
+ B = { B_I }
+ C = { C_I }
+ D = { D_I }
+ IEN = { CLRBAR & LOADBAR & ENP & ENT }
+ ILD = { CLRBAR & ~LOADBAR }
+ ILC = { ~(~CLRBAR | ILD) }
+ IA1 = { ILC & QA & ~IEN }
+ IA2 = { IEN & QABAR }
+ IB1 = { QB & ~(QA & IEN) & ILC }
+ IB2 = { QA & IEN & QBBAR }
+ IC1 = { QC & ~(QB & QA & IEN) & ILC }
+ IC2 = { QB & QA & IEN & QCBAR }
+ IDB = { QD & ~(QC & QB & QA & IEN) & ILC }
+ IDC = { QC & QB & QA & IEN & QDBAR }
+ RCO = { QD & QC & QB & QA & ENT }
+ DA = { IA1 | IA2 | (ILD & A) }
+ DB = { IB1 | IB2 | (ILD & B) }
+ DC = { IC1 | IC2 | (ILD & C) }
+ DD = { IDB | IDC | (ILD & D) }
*
U1S163ADLY PINDLY (5,0,10) DPWR DGND
+ RCO QA QB QC QD
+ CLK ENT ENP CLRBAR LOADBAR A B C D IEN
+ RCO_O QA_O QB_O QC_O QD_O
+ IO_LS MNTYMXDLY={MNTYMXDLY} IO_LEVEL={IO_LEVEL}
+ BOOLEAN:
+ CLOCK = { CHANGED_LH(CLK,0) }
+ PINDLY:
+ QA_O QB_O QC_O QD_O = {
+   CASE(
+     CLOCK & TRN_LH, DELAY(-1,13NS,24NS),
+     CLOCK & TRN_HL, DELAY(-1,18NS,27NS),
+     DELAY(-1,18NS,27NS)
+   )
+ }
+ RCO_O = {
+   CASE(
+     CHANGED(ENT,0), DELAY(-1,9NS,14NS),
```

```

+   CLOCK & TRN_HL, DELAY(-1,18NS,35NS),
+   CLOCK & TRN_LH, DELAY(-1,20NS,35NS),
+   DELAY(-1,20NS,35NS)
+   )
+ }
+ BOOLEAN:
+ NOTCLEAR = { CLRBAR!='0 ^ CHANGED(CLRBAR,0) }
+ FREQ:
+ NODE = CLK
+ MAXFREQ = 25MEG
+ WIDTH:
+ NODE = CLK
+ MIN_LO = 25NS
+ MIN_HI = 25NS
+ SETUP_HOLD:
+ DATA(4) = A B C D
+ CLOCK LH = CLK
+ SETUP_TIME = 20NS
+ HOLDTIME = 3NS
+ WHEN = { (LOADBAR!='1 ^ CHANGED(LOADBAR,0)) & NOTCLEAR }
+ SETUP_HOLD:
+ DATA(2) = ENP ENT
+ CLOCK LH = CLK
+ SETUP_TIME = 20NS
+ HOLDTIME = 3NS
+ WHEN = { (LOADBAR!='0 ^ CHANGED(LOADBAR,0)) & NOTCLEAR & CHANGED(IEN,20NS) }
+ SETUP_HOLD:
+ DATA(1) = LOADBAR
+ CLOCK LH = CLK
+ SETUP_TIME = 20NS
+ HOLDTIME = 3NS
+ WHEN = { NOTCLEAR }
+ SETUP_HOLD:
+ DATA(1) = CLRBAR
+ CLOCK LH = CLK
+ SETUP_TIME_LO = 20NS
+ SETUP_TIME_HI = 25NS
+ HOLDTIME = 3NS
+
*
.ENDS

```

Фиг. 7. Макромодел на 4 битов брояч 74LS163

4.3. Общ формат за обръщение към макромодел

Общият формат за обръщение към макромодел започва с означението на макромодела, следва изброяването на възлите, в които са включени изводите на компонента, и завършва с името на макромодела.

Xxxx <списък от възли><име на макромодела>

Пример за обръщение към макромодел

Означение: - **X_U3**;

Списък от възли: - CLOCK N35977 N35977 POR \$D_HI \$D_HI \$D_HI \$D_HI \$D_HI M_UN0005
M_UN0006 M_UN0007 M_UN0008 N35993 \$G_DPWR \$G_DGND;

Име на макромодела - **74LS163A** (синхронен четирибитов брояч);

Параметри на макромодела - IO_LEVEL=0 MNTYMXDLY=0

```

X_U3      CLOCK N35977 N35977 POR $D_HI $D_HI $D_HI $D_HI $D_HI M_UN0005
+ M_UN0006 M_UN0007 M_UN0008 N35993 $G_DPWR $G_DGND 74LS163A
+ PARAMS: IO_LEVEL=0 MNTYMXDLY=0

```